

# NAG Fortran Library Routine Document

## G13AAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G13AAF carries out non-seasonal and seasonal differencing on a time series. Information which allows the original series to be reconstituted from the differenced series is also produced. This information is required in time series forecasting.

### 2 Specification

```
SUBROUTINE G13AAF(X, NX, ND, NDS, NS, XD, NXD, IFAIL)
INTEGER          NX, ND, NDS, NS, NXD, IFAIL
real           X(NX), XD(NX)
```

### 3 Description

Let  $\nabla^d \nabla_s^D x_i$  be the  $i$ th value of a time series  $x_i$ , for  $i = 1, 2, \dots, n$  after non-seasonal differencing of order  $d$  and seasonal differencing of order  $D$  (with period or seasonality  $s$ ). In general,

$$\begin{aligned} \nabla^d \nabla_s^D x_i &= \nabla^{d-1} \nabla_s^D x_{i+1} - \nabla^{d-1} \nabla_s^D x_i & d > 0 \\ \nabla^d \nabla_s^D x_i &= \nabla^d \nabla_s^{D-1} x_{i+s} - \nabla^d \nabla_s^{D-1} x_i & D > 0 \end{aligned}$$

Non-seasonal differencing up to the required order  $d$  is obtained using

$$\begin{aligned} \nabla^1 x_i &= x_{i+1} - x_i & \text{for } i = 1, 2, \dots, (n-1) \\ \nabla^2 x_i &= \nabla^1 x_{i+1} - \nabla^1 x_i & \text{for } i = 1, 2, \dots, (n-2) \\ &\vdots \\ \nabla^d x_i &= \nabla^{d-1} x_{i+1} - \nabla^{d-1} x_i & \text{for } i = 1, 2, \dots, (n-d) \end{aligned}$$

Seasonal differencing up to the required order  $D$  is then obtained using

$$\begin{aligned} \nabla^d \nabla_s^1 x_i &= \nabla^d x_{i+s} - \nabla^d x_i & \text{for } i = 1, 2, \dots, (n-d-s) \\ \nabla^d \nabla_s^2 x_i &= \nabla^d \nabla_s^1 x_{i+s} - \nabla^d \nabla_s^1 x_i & \text{for } i = 1, 2, \dots, (n-d-2s) \\ &\vdots \\ \nabla^d \nabla_s^D x_i &= \nabla^d \nabla_s^{D+1} x_{i+s} - \nabla^d \nabla_s^{D+1} x_i & \text{for } i = 1, 2, \dots, (n-d-D \times s) \end{aligned}$$

Mathematically, the sequence in which the differencing operations are performed does not affect the final resulting series of  $m = n - d - D \times s$  values.

### 4 References

None.

### 5 Parameters

- 1: X(NX) – *real* array *Input*  
*On entry:* the undifferenced time series,  $x_i$ , for  $i = 1, 2, \dots, n$ .

- 2: NX – INTEGER *Input*  
*On entry:* the number of values,  $n$ , in the undifferenced time series.  
*Constraint:*  $NX > ND + (NDS \times NS)$ .
- 3: ND – INTEGER *Input*  
*On entry:* the order of non-seasonal differencing,  $d$ .  
*Constraint:*  $ND \geq 0$ .
- 4: NDS – INTEGER *Input*  
*On entry:* the order of seasonal differencing,  $D$ .  
*Constraint:*  $NDS \geq 0$ .
- 5: NS – INTEGER *Input*  
*On entry:* the seasonality,  $s$ .  
*Constraints:*  
     if  $NDS > 0$  then  $NS > 0$ ;  
     if  $NDS = 0$  then  $NS \geq 0$ .
- 6: XD(NX) – *real* array *Output*  
*On exit:* the differenced values in elements 1 to NXD, and reconstitution data in the remainder of the array.
- 7: NXD – INTEGER *Output*  
*On exit:* the number of differenced values in the array XD.
- 8: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, ND < 0,  
 or NDS < 0,  
 or NS < 0,  
 or NS = 0 when NDS > 0.

IFAIL = 2

On entry,  $NX \leq ND + (NDS \times NS)$ .

## 7 Accuracy

The computations are believed to be stable.

## 8 Further Comments

The time taken by the routine is approximately proportional to  $(ND + NDS) \times NX$ .

## 9 Example

The example program reads in a set of data consisting of 20 observations from a time series. Non-seasonal differencing of order 2 and seasonal differencing of order 1 (with seasonality of 4) are applied to the input data, giving an output array holding 14 differenced values and 6 values which can be used to reconstitute the output array.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G13AAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NXMAX
      PARAMETER        (NXMAX=20)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, ND, NDS, NS, NX, NXD
*      .. Local Arrays ..
      real            X(NXMAX), XD(NXMAX)
*      .. External Subroutines ..
      EXTERNAL         G13AAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G13AAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX, ND, NDS, NS
      IF (NX.GT.0 .AND. NX.LE.NXMAX) THEN
          READ (NIN,*) (X(I),I=1,NX)
          WRITE (NOUT,*)
          WRITE (NOUT,99999) 'Non-seasonal differencing of order ', ND,
+ ' and seasonal differencing'
          WRITE (NOUT,99999) 'of order ', NDS, ' with seasonality ', NS,
+ ' are applied'
          IFAIL = 0
*
          CALL G13AAF(X,NX,ND,NDS,NS,XD,NXD,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,99998) 'The output array holds ', NX,
+ ' values, of which the first ', NXD,
+ ' are differenced values'
          WRITE (NOUT,*)
          WRITE (NOUT,99997) (XD(I),I=1,NX)
      END IF
      STOP
*
99999 FORMAT (1X,A,I1,A,I1,A)
99998 FORMAT (1X,A,I2,A,I2,A)
99997 FORMAT (1X,5F9.1)
      END
```

## 9.2 Program Data

G13AAF Example Program Data

```
20 2 1 4
120.0 108.0 98.0 118.0 135.0
131.0 118.0 125.0 121.0 100.0
82.0 82.0 89.0 88.0 86.0
96.0 108.0 110.0 99.0 105.0
```

## 9.3 Program Results

G13AAF Example Program Results

Non-seasonal differencing of order 2 and seasonal differencing  
of order 1 with seasonality 4 are applied

The output array holds 20 values, of which the first 14 are differenced values

```
-11.0   -10.0   -8.0    4.0    12.0
-2.0    18.0    9.0   -4.0   -6.0
-5.0    -2.0  -12.0    5.0    2.0
-10.0  -13.0   17.0    6.0  105.0
```

---